

# WRITE

This chapter covers the following topics:

- Function
- Syntax 1 - Dynamic Formatting
- Syntax 1 - Description
- Syntax 2 - Using Predefined Form/Map
- Syntax 2 - Description
- Examples

For an explanation of the symbols used in the syntax diagram, see *Syntax Symbols*.

Related Statements: AT END OF PAGE | AT TOP OF PAGE | CLOSE PRINTER | DEFINE PRINTER | DISPLAY | EJECT | FORMAT | NEWPAGE | PRINT | SKIP | SUSPEND IDENTICAL SUPPRESS | WRITE TITLE | WRITE TRAILER

Belongs to Function Group: *Creation of Output Reports*

---

## Function

The WRITE statement is used to produce output in free format.


The WRITE statement differs from the DISPLAY statement in the following respects:

- Line overflow is supported. If the line width is exceeded for a line, the next field (or text) is written on the next line. Fields or text elements are not split between lines.
- No default column headers are created. The length of the data determines the number of positions printed for each field.
- A range of values/occurrences for an array is output horizontally rather than vertically.

See also the following topics (in the *Programming Guide*):

- *Controlling Data Output*
- *Statements DISPLAY and WRITE*
- *Index Notation for Multiple-Value Fields and Periodic Groups*
- *Example of DISPLAY VERT with WRITE Statement*
- *Layout of an Output Page*

## Syntax 1 - Dynamic Formatting



For an explanation of the symbols used in the syntax diagram, see *Syntax Symbols*.

## Syntax 1 - Description

Operand Definition Table:

Operand	Possible Structure	Possible Formats	Referencing Permitted	Dynamic Definition
<i>operand1</i>	S A G N	A U N P I F B D T L G O	yes	no

Syntax Element Description:

<b>(<i>rep</i>)</b>	<p><b>Report Specification:</b></p> <p>The notation (<i>rep</i>) is used to specify the identification of the report if multiple reports are to be produced by the program.</p> <p>As report identification, a value in the range 0 - 31 or a logical name which has been assigned using the DEFINE PRINTER statement may be specified.</p> <p>If (<i>rep</i>) is not specified, the statement will apply to the first report (Report 0).</p> <p>If this printer file is defined to Natural as PC, the report will be downloaded to the PC, see <i>Example 5</i>.</p> <p>For information on how to control the format of an output report created with Natural, see <i>Controlling Data Output</i> (in the <i>Programming Guide</i>).</p>
---------------------	---

<b>NOTITLE</b>	<p><b>Default Page Title Suppression:</b></p> <p>Natural generates a single title line for each page resulting from a WRITE statement. This title contains the page number, the time of day, and the date. Time of day is set at the beginning of program execution. This default title line may be overridden by using a WRITE TITLE statement, or it may be suppressed by using the NOTITLE option in the WRITE statement.</p> <p>Examples:</p> <ul style="list-style-type: none"> <li>● Default title will be produced: <code>WRITE NAME</code></li> <li>● User title will be produced: <code>WRITE NAME WRITE TITLE 'user-title'</code></li> <li>● No title will be produced: <code>WRITE NOTITLE NAME</code></li> </ul> <p><b>Notes:</b></p> <ol style="list-style-type: none"> <li>1. If the NOTITLE option is used, it applies to all DISPLAY, PRINT and WRITE statements within the same object which write data to the same report.</li> <li>2. Page overflow is checked <i>before</i> execution of a WRITE statement. No new page with title or trailer information is generated <i>during</i> the execution of a WRITE statement.</li> </ol>
<b>NOHDR</b>	<p><b>Column Header Suppression:</b></p> <p>The WRITE statement itself does not produce any column headers. However, if you use the WRITE statement in conjunction with a DISPLAY statement, you can use the NOHDR option of the WRITE statement to suppress the column headers generated by the DISPLAY statement. The NOHDR option only takes effect if the execution of the WRITE statement causes a new page to be output.</p> <p>Without the NOHDR option, the column headers (if any) of the DISPLAY statement would be output on this new page; with NOHDR they will not.</p>

<i>statement-parameters</i>	<p><b>Parameter Definition at Statement Level:</b></p> <p>One or more parameters, enclosed within parentheses, may be specified at statement level, that is, immediately after the WRITE statement.</p> <p>Each parameter specified will override the corresponding parameter previously specified in a GLOBALS command, SET GLOBALS (in Reporting Mode only) or FORMAT statement.</p> <p>If more than one parameter is specified, they must be separated by one or more blanks from one another. Each parameter specification must not be split between two statement lines.</p> <p><b>Note:</b> The parameter settings applied here will only be regarded for variable fields, but they have no effect on text-constants. If you would like to set field attributes for a text-constant, they have to be set explicitly for this element, see <i>Parameter Definition at Element (Field) Level</i>.</p> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>List of Parameters</i></li> <li>● <i>Example of Parameter Usage at Statement and Element (Field) Level</i></li> <li>● <i>Example 5 - WRITE Statement Using '=' and Parameters on Statement/Element (Field) Level</i></li> </ul>
<i>nX, nT, x/y, T*field-name, P*field-name, '=' , /,</i>	<p><b>Field Positioning Notation:</b></p> <p>See <i>Field Positioning Notations</i> in the section <i>Output Format Definitions</i>.</p>
<i>'text', 'c'(n), attributes, operand1, parameters</i>	<p><b>Text/Attribute Assignment:</b></p> <p>See <i>Text/Attribute Assignment</i> in the section <i>Output Format Definitions</i>.</p>

## List of Parameters

Parameters that can be specified with the WRITE statement		Specification (S = at statement level, E = at element level)
AD	Attribute Definition	SE
AL	Alphanumeric Length for Output	SE
BX	Box Definition	SE
CD	Color Definition	SE
CV	Control Variable	SE
DF	Date Format	SE
DL	Display Length for Output	SE
DY	Dynamic Attributes	SE
EM	Edit Mask	SE
FL	Floating Point Mantissa Length	SE
IS	Identical Suppress	SE
LS	Line Size	S
MC	Multiple-Value Field Count	S
MP	Maximum Number of Pages of a Report	S
NL	Numeric Length for Output	SE
PC	Periodic Group Count	S
PM	Print Mode	SE
PS	Page Size *	S
SG	Sign Position	SE
UC	Underlining Character	S
ZP	Zero Printing	SE

\* If the number of occurrences of an array exceeds the PS value, a NAT0303 error will be output.

The individual session parameters are described in the *Parameter Reference*.

See also the following topics in the *Programming Guide*:

- *Centering of Column Headers - HC Parameter*
- *Width of Column Headers - HW Parameter*
- *Filler Characters for Headers - Parameters FC and GC*
- *Underlining Character for Titles and Headers - UC Parameter*

## Example of Parameter Usage at Statement and Element (Field) Level


```

DEFINE DATA LOCAL
1 VARI (A4)      INIT <'1234'>          /*      Output
END-DEFINE      /*      Produced
*              /*      -----
WRITE          'Text'                 VARI      /*      Text 1234
WRITE (PM=I)  'Text'                 VARI      /*      Text 4321
WRITE          'Text' (PM=I)          VARI (PM=I) /*      txeT 4321
WRITE          'Text' (PM=I)          VARI      /*      txeT 1234
END

```

See also *Example 5 - WRITE Statement Using '=' and Parameters on Statement/Element (Field) Level*.

## Output Format Definitions



For an explanation of the symbols used in the syntax diagram, see *Syntax Symbols*.

## Field Positioning Notations

<b><math>n\mathbf{X}</math></b>	<p><b>Column Spacing:</b></p> <p>This notation inserts <math>n</math> spaces between columns. <math>n</math> must not be zero.</p> <p>Example:</p> <pre>WRITE NAME 5X SALARY</pre> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>Example 2 - WRITE Statement Using <math>n\mathbf{X}</math>, <math>n\mathbf{T}</math> Notation (below)</i></li> <li>● <i>Column Spacing - SF Parameter and <math>n\mathbf{X}</math> Notation (in the Programming Guide)</i></li> </ul>
---------------------------------	---

<b><i>nT</i></b>	<p><b>Tab Setting:</b></p> <p>The <i>nT</i> notation causes positioning (tabulation) to print position <i>n</i>. Backward positioning is not permitted.</p> <p>In the following example, NAME is printed beginning in position 25, and SALARY is printed beginning in position 50:</p> <pre>WRITE 25T NAME 50T SALARY</pre> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>Example 2 - WRITE Statement Using nX, nT Notation</i> (below)</li> <li>● <i>Tab Setting - nT Notation</i> (in the <i>Programming Guide</i>)</li> </ul>
<b><i>x/y</i></b>	<p><b><i>x/y</i> Positioning:</b></p> <p>The <i>x/y</i> notation causes the next element to be placed <i>x</i> lines below the output of the last statement, beginning in column <i>y</i>. <i>y</i> must not be zero. Backward positioning in the same line is not permitted.</p> <p>See also <i>Positioning Notation x/y</i> (in the <i>Programming Guide</i>).</p>
<b><i>T*field-name</i></b>	<p><b>Field Related Positioning:</b></p> <p>The notation <i>T*</i> is used to position to a <i>specific print position of a field</i> used in a previous DISPLAY statement. Backward positioning is not permitted.</p> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>Example 3 - WRITE Statement Using T* Notation</i> (below)</li> <li>● <i>Tab Notation - T*field</i> (in the <i>Programming Guide</i>)</li> </ul>
<b><i>P*field-name</i></b>	<p><b>Field and Line Related Positioning:</b></p> <p>The notation <i>P*</i> is used to position to a <i>specific print position and line of a field</i> used in a previous DISPLAY statement. It is most often used in conjunction with vertical printing mode. Backward positioning is not permitted.</p> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>Example 4 - WRITE Statement Using P* Notation</i> (below)</li> <li>● <i>Tab Notation P*field</i> (in the <i>Programming Guide</i>)</li> </ul>

'='	<p><b>Field Content Positioned behind Field Heading:</b></p> <p>When placed before a field, the equal sign '=' results in the display of the field heading (as defined in the DEFINE DATA statement or in the DDM) followed by the field contents.</p> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>Example 1 - WRITE Statement Using '=', 'text', '/'</i></li> <li>● <i>Example 5 - WRITE Statement Using '=' and Statement/Element Parameters</i></li> </ul>
/	<p><b>Line Advance - Slash Notation:</b></p> <p>When placed between fields or text elements, a slash (/) causes positioning to the beginning of the next print line.</p> <p>Example:</p> <pre>WRITE NAME / SALARY</pre> <p>Multiple slash (/) notations may be used to cause multiple line advances.</p> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>Example 1 - WRITE Statement Using '=', 'text', '/' (below)</i></li> <li>● <i>Line Advance - Slash Notation (in the Programming Guide)</i></li> <li>● <i>Example 2 - Line Advance in WRITE Statement (in the Programming Guide)</i></li> </ul>

### Text/Attribute Assignment


'text'	<p><b>Text Assignment:</b></p> <p>The character string enclosed by single quotes is displayed.</p> <p>Example:</p> <pre>WRITE 'EMPLOYEE' NAME 'MARITAL/STATUS' MAR-STAT</pre> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>Example 1 - WRITE Statement Using '=', 'text', '/' (below)</i></li> <li>● <i>Text Notation, Defining a Text to Be Used with a Statement (in the Programming Guide)</i></li> </ul>
--------	--



'c'(n)	<p><b>Character Repetition:</b></p> <p>The character enclosed by single quotes is displayed <i>n</i> times immediately before the field value.</p> <p>For example:</p> <pre>WRITE '*' (5) '=' NAME</pre> <p>results in</p> <pre>***** SMITH</pre> <p>See also <i>Text Notation, Defining a Character to Be Displayed n Times before a Field Value</i> (in the <i>Programming Guide</i>).</p>
attributes	<p><b>Field Representation and Color Attributes:</b></p> <p>It is possible to assign various attributes for text/field display. These attributes and the syntax that may be used are described in the section <i>Output Attributes</i> below.</p> <p>Examples:</p> <pre>WRITE 'TEXT' (BGR) WRITE 'TEXT' (B) WRITE 'TEXT' (BBLC)</pre>
operand1	<p><b>Field to be Written:</b></p> <p><i>operand1</i> specifies the field whose content is to be written in this place.</p> <p>Arrays with ranges that allow to vary the number of occurrences at execution time may not be specified.</p> <p><b>Note:</b> For DL/I databases: The DL/I AIX fields can be displayed only if a PCB is used with the AIX specified in the parameter PROCSEQ. If not, an error message is returned by Natural at runtime.</p>
parameters	<p><b>Parameter Definition at Element (Field) Level:</b></p> <p>One or more parameters, enclosed within parentheses, may be specified at element (field) level, that is, immediately after <i>operand1</i>. Each parameter specified in this manner will override the corresponding parameter previously specified at statement level or in a GLOBALS command, SET GLOBALS (in Reporting Mode only) or FORMAT statement.</p> <p>If more than one parameter is specified, one or more blanks must be placed between each entry. An entry may not be split between two statement lines.</p> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>List of Parameters</i></li> <li>● <i>Example of Parameter Usage at Statement and Element (Field) Level</i></li> </ul>

### Output Attributes

*attributes* indicates the output attributes to be used for text display. Attributes may be:




For the possible session parameter values, refer to the corresponding sections in the *Parameter Reference* documentation:

- *AD - Attribute Definition*, section *Field Representation*
- *CD - Color Definition*
- *BX - Box Definition*
- *PM - Print Mode*

**Note:**

The compiler actually accepts more than one attribute value for an output field. For example, you may specify: AD=BDI. In such a case, however, only the last value applies. In the given example, only the value **I** will become effective and the output field will be displayed intensified.

## Syntax 2 - Using Predefined Form/Map



For an explanation of the symbols used in the syntax diagram, see *Syntax Symbols*.

## Syntax 2 - Description

Operand Definition Table:

Operand	Possible Structure		Possible Formats													Referencing Permitted	Dynamic Definition			
<i>operand1</i>	C	S				A													no	no
<i>operand2</i>		S	A	G	N	A	U	N	P	I	F	B	D	T	L				yes	no

## Syntax Element Description:

<b>[USING] FORM [USING] MAP</b>	<p><b>Use of Predefined Form/Map Layout:</b></p> <p>This option may be used to indicate that a form/map layout previously defined using the Natural map editor is to be used.</p> <p>A map layout used in a WRITE statement does not automatically create a new page each time the map is output.</p> <p>For the line spacing, the LS parameter setting must be 1 byte greater than the LS setting defined in the map.</p>
<i>operand1</i>	<p><b>Form/Map Name:</b></p> <p><i>operand1</i> is the name of the form/map to be used.</p>
<i>operand2</i>	<p><b>Field to be Written:</b></p> <p><i>operand2</i> is the name(s) of the field(s) to be written.</p> <p>If <i>operand1</i> is a constant and <i>operand2</i> is omitted, the fields are taken from the map source at compilation time.</p> <p>The fields must agree in number, sequence, format, length and (for arrays) number of occurrences with the fields in the referenced form/map; otherwise, an error occurs.</p>
<b>NOTITLE/NOHDR</b>	<p><b>Title Line/Column Header Suppression:</b></p> <p>NOTITLE and NOHDR are described under <i>Syntax 1</i> of the WRITE statement.</p>

## Examples

- Example 1 - WRITE Statement Using '=', 'text', '/'
- Example 2 - WRITE Statement Using nX, nT Notation
- Example 3 - WRITE Statement Using T\* Notation
- Example 4 - WRITE Statement Using P\* Notation
- Example 5 - WRITE Statement Using '=' and Parameters on Statement/Element (Field) Level
- Example 6 - Report Specification with Output File Defined to Natural as PC

### Example 1 - WRITE Statement Using '=', 'text', '/'

```
** Example 'WRTEX1': WRITE (with '=', 'text', '/')
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 FULL-NAME
    3 FIRST-NAME
```

```

      3 MIDDLE-I
      3 NAME
      2 CITY
      2 COUNTRY
END-DEFINE
*
LIMIT 1
READ EMPL-VIEW BY NAME
/*
  WRITE NOTITLE
    '=' NAME '=' FIRST-NAME '=' MIDDLE-I //
    'L O C A T I O N' /
    'CITY:'  ' CITY      /
    'COUNTRY:' COUNTRY //
/*
END-READ
END

```

### Output of Program WRTEX1:

```

NAME: ABELLAN           FIRST-NAME: KEPA           MIDDLE-I:

L O C A T I O N
CITY:  MADRID
COUNTRY: E

```

### Example 2 - WRITE Statement Using nX, nT Notation

```

** Example 'WRTEX2': WRITE (with nX, nT notation)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 JOB-TITLE
END-DEFINE
*
LIMIT 4
READ EMPL-VIEW BY NAME
  WRITE NOTITLE 5X NAME 50T JOB-TITLE
END-READ
END

```

### Output of WRTEX2:

```

ABELLAN           MAQUINISTA
ACHIESON          DATA BASE ADMINISTRATOR
ADAM              CHEF DE SERVICE
ADKINSON          PROGRAMMER

```

### Example 3 - WRITE Statement Using T\* Notation

```

** Example 'WRTEX3': WRITE (with T* notation)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 CITY
  2 SALARY (1)
END-DEFINE
*
LIMIT 5

```

```

READ EMPL-VIEW BY CITY STARTING FROM 'ALBU'
  DISPLAY NOTITLE CITY NAME SALARY (1)
  AT BREAK CITY
  /*
  WRITE / 'CITY AVERAGE:' T*SALARY (1) AVER(SALARY(1)) //
  /*
END-BREAK
END-READ
END
    
```

**Output of Program WRTEX3:**

CITY	NAME	ANNUAL SALARY
ALBUQUERQUE	HAMMOND	22000
ALBUQUERQUE	ROLLING	34000
ALBUQUERQUE	FREEMAN	34000
ALBUQUERQUE	LINCOLN	41000
CITY AVERAGE:		32750
ALFRETON	GOLDBERG	4800
CITY AVERAGE:		4800

**Example 4 - WRITE Statement Using P\* Notation**

```

** Example 'WRTEX4': WRITE (with P* notation)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 CITY
  2 BIRTH
  2 SALARY (1)
END-DEFINE
*
LIMIT 3
READ EMPL-VIEW BY CITY FROM 'N'
  DISPLAY NOTITLE NAME CITY
  VERT AS 'BIRTH/SALARY' BIRTH (EM=YYYY-MM-DD) SALARY (1)
  SKIP 1
  AT BREAK CITY
  WRITE / 'CITY AVERAGE' P*SALARY (1) AVER(SALARY (1)) //
  END-BREAK
END-READ
END
    
```

**Output of Program WRTEX4:**

NAME	CITY	BIRTH SALARY
WILCOX	NASHVILLE	1970-01-01 38000
MORRISON	NASHVILLE	1949-07-10 36000

```

CITY AVERAGE                                37000

BOYER                NEMOURS                1955-11-23
                                      195900

CITY AVERAGE                                195900

```

## Example 5 - WRITE Statement Using '=' and Parameters on Statement/Element (Field) Level

```

** Example 'WRTEX5': WRITE (using '=', statement/element parameters)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 PERSONNEL-ID
  2 PHONE
END-DEFINE
*
LIMIT 2
READ EMPL-VIEW BY NAME
  WRITE NOTITLE (AL=16 NL=8)
    '=' PERSONNEL-ID '=' NAME '=' PHONE (AL=10 EM=XXX-XXXXXXX)
END-READ
END

```

### Output of Program WRTEX5:

```

PERSONNEL ID: 60008339      NAME: ABELLAN      TELEPHONE: 435-6726
PERSONNEL ID: 30000231      NAME: ACHIESON     TELEPHONE: 523-341

```

## Example 6 - Report Specification with Output File Defined to Natural as PC

```

** Example 'PCDIEX1': DISPLAY and WRITE to PC
**
** NOTE: Example requires that Natural Connection is installed.
*****
DEFINE DATA LOCAL
01 PERS VIEW OF EMPLOYEES
  02 PERSONNEL-ID
  02 NAME
  02 CITY
END-DEFINE
*
FIND PERS WITH CITY = 'NEW YORK'          /* Data selection
  WRITE (7) TITLE LEFT 'List of employees in New York' /
  DISPLAY (7)          /* (7) designates the output file (here the PC).
    'Location'  CITY
    'Surname'   NAME
    'ID'        PERSONNEL-ID
END-FIND
END

```